



# Глава 15

## Формулы и функции для финансовых расчетов

### В ЭТОЙ ГЛАВЕ...

- Стоимость денег во времени
- Кредитный калькулятор
- Расчет параметров инвестиций
- Расчет амортизации
- Финансовое прогнозирование

**Н**е секрет, что чаще всего Excel используется для всевозможных финансовых расчетов. Каждый день люди принимают тысячи решений, основываясь на результатах вычислений в электронных таблицах. Принимаемые решения могут быть как очень простыми (“могу ли я купить этот автомобиль?”), так и достаточно сложными (“окупятся ли инвестиции в компанию ABC за 18 месяцев?”). В настоящей главе описаны основные приемы, применяемые в Excel для финансовых расчетов.

### Стоимость денег во времени

Номинальная стоимость денег не всегда может соответствовать фактической. Ключевое понятие здесь — *стоимость денег во времени (СДВ)*. Вычисления, выполняемые над денежными суммами, могут производиться в прошлом, настоящем или в будущем времени. Выбор типа вычислений основывается на том, что сумма денег увеличивается, поскольку постоянно поступают начисления по процентам. Другими словами, вложив один рубль сегодня, завтра вы получите больше.

Например, предположим, что богатенький заокеанский дядюшка решил подарить вам деньги, причем вы должны выбрать один из перечисленных ниже вариантов.

- Получить 800 000 Р сегодня.
- Получить 950 000 Р через год.
- Получить 1 200 000 Р через пять лет.
- Получать 15 000 р. каждый месяц на протяжении пяти лет.

Чтобы получить максимальную выгоду, нужно учитывать не только собственно номинальную сумму, но и ту *стоимость*, которую будут представлять деньги на момент, когда вы их получите на руки.

Стоимость денег во времени зависит от ваших возможностей и перспектив. Другими словами, вы можете быть либо кредитором, либо заемщиком. Когда вы берете кредит на покупку автомобиля, вы выступаете в роли заемщика, а учреждение, которое дает вам деньги, — в роли кредитора. Когда же вы вкладываете деньги в банк на депозит (сберегательный счет), вы — кредитор, поскольку занимаете свои деньги банку, а банк является вашим заемщиком.

Со *стоимостью денег во времени* связано еще несколько понятий.


- **Приведенная (текущая) стоимость.** Это *основная* сумма, или *капитал*, на который начисляются проценты. Например, если на депозит в банке вкладывается 500 000 Р, то эта величина представляет собой капитал или приведенную (текущую) стоимость вложенных денег. Если берется ссуда размером 1 500 000 Р. на приобретение автомобиля, то данная сумма будет основной или приведенной стоимостью ссуды. Приведенная стоимость может быть как положительной, так и отрицательной.
- **Будущая стоимость.** Представляет собой сумму приведенной (текущей) стоимости и начисленных по ней процентов. Например, если на депозитный счет в банке вкладывается 50 000 Р на пять лет под 6% годовых, то в конце срока можно будет получить 66 911,28 Р. Последняя сумма будет будущей стоимостью инвестиции. Если же берется ссуда на три года на покупку автомобиля в размере 150 000 Р. под 7% годовых, то в конце срока нужно будет выплатить 240 867,22 Р. Иными словами, нужно будет вернуть основную сумму плюс проценты. Будущая стоимость, в зависимости от перспективы (кредитор или заемщик), может быть положительной или отрицательной.
- **Взнос.** Взносом (платежом) может быть либо капитал, либо капитал и начисленные на него проценты. Если каждый месяц вкладывается по 1000 Р на депозитный счет, то 1000 Р — это взнос. Если для погашения ссуды ежемесячный взнос составляет 8 250 Р, то он состоит из тела кредита (основной суммы) и начисленных процентов.
- **Процентная ставка.** Часть основной суммы (в процентах), начисляемая за определенный период (как правило, за год). Например, деньги могут быть вложены в банк на депозит с процентной ставкой 5,5% годовых. Или процентная ставка по кредиту может составлять 7,75% в год.
- **Период.** Промежуток времени, по истечении которого выплачиваются проценты (например, банк может выплачивать проценты по депозитам ежеквартально, тогда как кредит за купленный автомобиль чаще всего приходится гасить ежемесячно).
- **Срок.** Промежуток времени, на который вкладываются или берутся в кредит деньги. Например, деньги вкладываются в банк на депозит на срок один год или ссуда берется на срок 30 лет.

## Кредитный калькулятор

Теперь рассмотрим, как рассчитываются выплаты по кредиту. Следует помнить, что кредит состоит из перечисленных ниже компонентов.

- Тело кредита (собственно размер ссуды, которую вы взяли в банке)
- Процентная ставка
- Сроки оплаты
- Размер регулярного платежа

Если известны любые три параметра, то можно создать формулу для вычисления четвертого параметра.



Все вычисления в этом разделе выполняются для ссуды с фиксированными процентной ставкой и сроком.

### Функции для расчета выплат по кредиту

В настоящем разделе описываются шесть функций: ПЛТ, ОСПЛТ, ПРПЛТ, СТАВКА, КПЕР и ПС. В табл. 15.1 описаны аргументы этих функций.

**Таблица 15.1. Аргументы функций для финансовых расчетов**

Аргумент	Описание
<i>Ставка</i>	Процентная ставка за один период. Если она выражена в процентах за год, то эту величину нужно разделить на количество периодов (месяцев)
<i>Кпер</i>	Общее количество выплат (периодов)
<i>Период</i>	Заданный период, который должен быть меньше или равен значению параметра <i>Кпер</i>
<i>Пс</i>	Приведенная стоимость, или основная сумма
<i>Плт</i>	Сумма, которая вносится за каждый период (постоянная величина)
<i>Бс</i>	Необязательный аргумент, равный будущей стоимости после последней выплаты. Если аргумент опущен, то он полагается равным нулю. (Например, будущая стоимость ссуды после ее погашения, очевидно, равна нулю.)
<i>Тип</i>	Указывает, когда должны осуществляться платежи. Равен нулю, если платежи выполняются в конце периода, и единице – если в начале. Это необязательный аргумент, который по умолчанию равен нулю
<i>Оценка</i>	Используется в функции СТАВКА и означает начальную оценку ожидаемого результата. Функция СТАВКА выполняет пошаговое вычисление. Поэтому, если полученный результат не будет сходиться, следует изменить значение этого аргумента

### Функция ПЛТ

Функция ПЛТ позволяет рассчитать размер выплат по кредиту (основная сумма плюс проценты) за один период, полагая постоянными размер выплат и процентную ставку. Эта функция имеет такой синтаксис:

ПЛТ (*Ставка*; *Кпер*; *Пс*; *Бс*; *Тип*)

Следующая формула возвращает размер ежемесячных выплат по кредиту, размер которого 50 000 Р, а процентная ставка составляет 6% годовых; кредит взят на срок четыре года (48 месяцев):

=ПЛТ(6%/12;48;-50000)

Вычислив формулу, получим, что размер ежемесячных выплат по ссуде равен 1174,25 Р. В качестве первого аргумента *Ставка* указан размер годовой ставки по кредиту, деленный на 12 месяцев (количество месяцев в году). Обратите также внимание на то, что третий аргумент (*Пс*, приведенная стоимость) — отрицательный. Это указывает на то, что деньги были взяты в долг.

### Функция ОСПЛТ


Функция **ОСПЛТ** возвращает *основную* часть платежа по кредиту за определенный период, подразумевая постоянным размер выплат и фиксированную процентную ставку. Функция имеет следующий синтаксис:

ОСПЛТ (*Ставка*; *Период*; *Кпер*; *Пс*; *Бс*; *Тип*)

Следующая формула возвращает основную часть выплаты за первый месяц по ссуде 50 000 Р, взятой под 6% годовых; ссуда взята на срок четыре года (48 месяцев):

=ОСПЛТ(6%/12;1;48;-50000)

Вычислив формулу, получим, что основная часть первой выплаты равна 924,25 Р, или приблизительно 78,7% от общей суммы выплаты. Если в качестве второго аргумента ввести число 48 (для вычисления основной части последней выплаты), то формула вернет 1168,41 Р, т.е. приблизительно 99,5% от общей суммы выплаты.



Чтобы вычислить размер основного кумулятивного платежа, который нужно будет сделать между любыми двумя определенными периодами, можно воспользоваться функцией **ОБЩДОХОД**. Функция содержит два дополнительных аргумента: *нач\_период* и *кон\_период*. В версиях, предшествовавших Excel 2007, функция **ОБЩДОХОД** становилась доступной только после установки надстройки **Пакет анализа**.

### Функция ПРПЛТ


Функция **ПРПЛТ** вычисляет ту часть общей суммы выплат по кредиту, которая идет на погашение процентов, полагая постоянными размер выплат и процентную ставку. Функция имеет следующий синтаксис:

ПРПЛТ (*Ставка*; *Период*; *Кпер*; *Пс*; *Бс*; *Тип*)

Формула, приведенная ниже, вычисляет размер первой выплаты по процентам для ссуды 50 000 Р, взятой под 6% годовых; ссуда взята на 4 года (48 месяцев):

=ПРПЛТ(6%/12;1;48;-50000)

После вычисления формулы получим, что сумма выплат по процентам за первый месяц составляет 250,00 Р, но размер выплат по процентам за последний месяц будет всего 5,84 Р.



Чтобы вычислить размер кумулятивного платежа по процентам, который нужно будет сделать между любыми двумя определенными периодами, следует воспользоваться функцией **ОБЩПЛАТ**. Функция содержит два дополнительных аргумента: *нач\_период* и *кон\_период*. В версиях, предшествовавших Excel 2007, функция **ОБЩПЛАТ** становилась доступной только после установки надстройки **Пакет анализа**.

## Функция СТАВКА

Функция СТАВКА позволяет вычислить размер процентной ставки по кредиту за один период на основании указанного общего количества периодов выплат, размера регулярного платежа и размера ссуды. Функция СТАВКА имеет следующий синтаксис:

СТАВКА (Кпер; Плт; Пс; Бс; Тип; Оценка)

Следующая формула вычисляет годовую процентную ставку для ссуды размером 50 000 Р, взятой на 48 месяцев; размер регулярного платежа составляет 1174,25 Р:

=СТАВКА(48;1174,25;-50000)\*12

Формула вернет 6,00%. Заметьте, что результат вычисления функции умножается на 12, поскольку функция возвращает процентную ставку за один период (месяц). Поэтому, чтобы получить годовую процентную ставку, нужно умножить полученное значение на количество периодов выплат в году (в нашем примере — на 12).

## Функция КПЕР

Функция КПЕР возвращает общее количество периодов выплат по ссуде по заданному размеру ссуды, процентной ставке и размеру регулярного платежа. Функция имеет следующий синтаксис:

КПЕР (Ставка; Плт; Пс; Бс; Тип)

Формула, приведенная ниже, вычисляет количество выплат по ссуде размером 50 000 Р и размеру ежемесячной выплаты, равному 1174,25 Р; ссуда взята под 6% годовых:

=КПЕР(6%/12;1174,25;-50000)

С небольшой погрешностью (меньше 0,0001) функция вернет число 48. Это значит, что всего нужно будет сделать 48 (за 48 месяцев) выплат по кредиту. Полученный результат оказался неточным из-за того, что размер выплат указан с точностью до одной копейки, т.е. округлен.

## Функция ПС

Функция ПС вычисляет приведенную стоимость ссуды (т.е. первоначальный размер кредита) по заданным процентной ставке, количеству периодов и размеру регулярного платежа. Функция имеет такой синтаксис:

ПС (Ставка; Кпер; Плт; Бс; Тип)

Следующая формула возвращает размер первоначальной ссуды, взятой на 48 месяцев под 6% годовых; размер регулярного платежа равен 1174,25 Р в месяц:

=ПС(6%/12;48;-1174,25)

Формула вернет 49 999,94 Р. Поскольку размер регулярного платежа указан с точностью до одной копейки, результат имеет погрешность в 6 копеек.

## Пример кредитного калькулятора

На рис. 15.1 показан рабочий лист, используя который, можно вычислить размер регулярного платежа по кредиту. В ячейке В1 находится сумма кредита, а в ячейке В2 — годовая процентная ставка. В ячейке В3 содержится длительность одного периода, выраженная в месяцах. Например, если в ячейке В3 находится число 1, значит, выплаты производятся ежемесячно. Общее количество периодов хранится в ячейке В4. В примере, показанном на рис. 15.1, вычисляется размер ежемесячной выплаты по кредиту размером 100 000 Р, взятому под 6,25% годовых на 36 месяцев. В ячейке В6 содержится следующая формула:

=ПЛТ(В2\*(В3/12);В4;-В1)

	A	B	C	D
1	Сумма кредита:	100 000,00 Р		
2	Годовая процентная ставка:	6,25%		
3	Периодичность платежей (мес.)	1		
4	Количество периодов:	36		
5				
6	Регулярный платеж:	3 053,53 Р		
7				
8				
9	Период:	36		
10	Тело кредита:	3 037,71 Р		
11	Проценты:	15,82 Р		
12				
13				

Рис. 15.1. Вычисление размера регулярного платежа по кредиту с помощью функции ПЛТ

Обратите внимание на то, что первым аргументом является выражение, которое вычисляет *периодическую процентную ставку* на основе значений годовой процентной ставки и периодичности выплат. Поэтому, если выплаты производятся ежеквартально на протяжении трех лет, периодичность выплат будет равна 3, а количество периодов — 12. Для того чтобы вычислить периодическую процентную ставку, нужно умножить годовую ставку на 3/12.



Рабочую книгу с примерами для данного раздела можно найти в файле loan payment.xlsx на сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется Pic15\_01.xlsx.

В ячейках В10 и В11 вычисляются размеры платежей, погашающих тело кредита и проценты за период, указанный в ячейке В9. Не забывайте, что значение в ячейке В9 должно быть меньше или равно значению ячейки В4. Иными словами, заданный период не должен превышать общее количество периодов.

Формула в ячейке В10 представлена ниже; она позволяет вычислить размер тела кредита для заданного в ячейке В9 периода:

=ОСПЛТ(В2\*(В3/12);В9;В4;-В1)

Приведенная ниже формула (ячейка В11) вычисляет размер выплат по процентам для заданного в ячейке В9 периода:

=ПРПЛТ (В2\* (В3/12) ;В9;В4; -В1)

Обратите внимание на то, что сумма ячеек В10 и В11 всегда равна размеру регулярного платежа по кредиту, который вычисляется в ячейке В6. Тем не менее соотношение между платежами по кредиту и выплатам по процентам зависит от текущего периода. (Доля платежей по кредиту с каждой последующей выплатой увеличивается, а выплат по процентам — уменьшается.) На рис. 15.2 графически показаны размеры выплачиваемых сумм для всего срока погашения кредита.




Рис. 15.2. Соотношение между платежами по кредиту и выплатам по процентам в зависимости от периода выплат

## Расчеты по кредитной карточке

Предположим, у вас есть кредитная карточка с определенной начальной суммой и вы хотите узнать, за сколько месяцев вы сможете погасить этот кредит при условии внесения минимально возможного ежемесячного платежа. На рис. 15.3 показан рабочий лист, на котором выполняются вычисления такого типа.

В диапазоне В1:В5 записаны исходные данные по кредитной карточке. В нашем примере начальная сумма на кредитной карточке составляет 10 000 Р, годовая процентная ставка равна 21,25%. Здесь принято, что минимальная сумма платежа составляет 2% от суммы, хранимой на кредитной карте, т.е. в данном случае минимальный платеж будет составлять 200 Р в месяц. В ячейку В5 вы можете ввести любую другую сумму, но она должна быть достаточной для погашения всей суммы кредита. Например, вы можете вносить ежемесячно по 500 Р, чтобы погасить кредит быстрее. С другой стороны, очевидно, что 100 Р в месяц будет недостаточно, поэтому формула возвратит ошибку.



Рабочую книгу с примерами для данного раздела можно найти в файле credit card payments.xlsx на сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется Pic15\_03.xlsx.

B7		
A	B	C
1	Начальная сумма:	10 000,00 Р
2	Годовая процентная ставка:	21,25%
3	Минимальный платеж в %:	2,00%
4	Минимальная сумма платежа в мес.:	200,00 Р
5	Реальный ежемесячный платеж:	200,00 Р
6		
7	Количество платежей:	123,4
8	Общая сумма:	24 684,19 Р
9	Из них проценты:	14 684,19 Р

**Рис. 15.3.** На этом рабочем листе подсчитывается количество платежей для погашения баланса по кредитной карте при условии внесения минимального ежемесячного платежа

В диапазоне B7:B9 выполняются вычисления. Приведенная ниже формула, записанная в ячейке B7, вычисляет, за сколько месяцев вы сможете погасить баланс по кредитной карте:

=КПЕР(B2/12;B5;-B1;0)

Здесь предполагается, что в конце вычисленного периода баланс по карте должен быть погашен, поэтому последний аргумент в функции КПЕР равен нулю. Простая формула в ячейке B8 вычисляет общую сумму, которую вы должны будете заплатить за все время с учетом начисленных процентов:

=B7\*B5

Формула в ячейке B9 вычисляет сумму, начисленную по процентам:

=B8-B1

В этом примере наглядно показано, что кредит в 10 000 Р вы будете гасить 123,4 месяца (т.е. более 10 лет!) при условии внесения ежемесячного минимального платежа в 200 Р. За это же время на взятые в кредит по карте 10 000 Р вам будет начислено 8932,93 Р процентов. Конечно, эти вычисления предполагают, что за все время пользования карточкой вы не будете снимать с нее никаких дополнительных средств. Из этого примера вы уже должны были понять, почему банки так упорно рассылают кредитные карты по обычной почте буквально направо и налево, а также всячески поощряют вас пользоваться такими картами.

На рис. 15.4 показаны дополнительные вычисления, связанные с нашим примером с кредитной карточкой. Например, если вы захотите погасить баланс по карточке за 12 месяцев, то каждый месяц вы должны будете вносить по 932,34 Р. При этом общая сумма выплат составит 11 188,06 Р, а начисления по процентам — 1 188,06 Р. В ячейке B13 записана такая формула:

=ПЛТ(\$B\$2/12;A13;-\$B\$1)

## Создание графика погашения ссуды

*График погашения ссуды* представляет собой таблицу, содержащую различную информацию по каждому периоду выплат. На рис. 15.5 показан рабочий лист, который содержит формулы, используемые для создания графика погашения ссуды.



	A	B	C	D
11				
12	<b>Периодичность выплат:</b>	<b>Требуемый платеж</b>	<b>Всего уплачено</b>	<b>Проценты</b>
13	2	5 133,20 Р	10 266,40 Р	266,40 Р
14	6	1 771,48 Р	10 628,86 Р	628,86 Р
15	12	932,34 Р	11 188,06 Р	1 188,06 Р
16	24	515,09 Р	12 362,05 Р	2 362,05 Р
17	36	378,04 Р	13 609,28 Р	3 609,28 Р
18	48	311,00 Р	14 928,15 Р	4 928,15 Р
19	60	271,94 Р	16 316,51 Р	6 316,51 Р
20	72	246,83 Р	17 771,67 Р	7 771,67 Р
21	84	229,65 Р	19 290,55 Р	9 290,55 Р
22	96	217,39 Р	20 869,69 Р	10 869,69 Р
23	108	208,38 Р	22 505,39 Р	12 505,39 Р
24	120	201,61 Р	24 193,77 Р	14 193,77 Р
25	132	196,45 Р	25 930,86 Р	15 930,86 Р

Рис. 15.4. В столбце В приведены размеры платежей, которые нужно сделать, чтобы погасить баланс по кредитной карте за указанный период

Рис15\_05.xlsx - Excel

Формула в B9:  $=ПЛТ(\$C\$2*(\$C\$3/12);\$C\$4;-\$C\$1)$


Номер платежа	Размер платежа	Всего выплачено	Выплаты по процентам	Сумма по процентам	Тело кредита	Сумма по кредиту	Остаток
		18 500,00 Р					18 500,00 Р
1	573,34 Р	573,34 Р	111,77 Р	111,77 Р	461,57 Р	461,57 Р	18 038,43 Р
2	573,34 Р	1 146,69 Р	108,98 Р	220,75 Р	464,36 Р	925,93 Р	17 574,07 Р
3	573,34 Р	1 720,03 Р	106,18 Р	326,93 Р	467,17 Р	1 393,10 Р	17 106,90 Р
4	573,34 Р	2 293,37 Р	103,35 Р	430,28 Р	469,99 Р	1 863,09 Р	16 636,91 Р
5	573,34 Р	2 866,72 Р	100,51 Р	530,80 Р	472,83 Р	2 335,92 Р	16 164,08 Р
6	573,34 Р	3 440,06 Р	97,66 Р	628,46 Р	475,69 Р	2 811,60 Р	15 688,40 Р
7	573,34 Р	4 013,40 Р	94,78 Р	723,24 Р	478,56 Р	3 290,16 Р	15 209,84 Р
8	573,34 Р	4 586,75 Р	91,89 Р	815,13 Р	481,45 Р	3 771,61 Р	14 728,39 Р
9	573,34 Р	5 160,09 Р	88,98 Р	904,12 Р	484,36 Р	4 255,97 Р	14 244,03 Р
10	573,34 Р	5 733,43 Р	86,06 Р	990,18 Р	487,29 Р	4 743,26 Р	13 756,74 Р
11	573,34 Р	6 306,78 Р	83,11 Р	1 073,29 Р	490,23 Р	5 233,49 Р	13 266,51 Р
12	573,34 Р	6 880,12 Р	80,15 Р	1 153,44 Р	493,19 Р	5 726,68 Р	12 773,32 Р
13	573,34 Р	7 453,46 Р	77,17 Р	1 230,61 Р	496,17 Р	6 222,85 Р	12 277,15 Р
14	573,34 Р	8 026,81 Р	74,17 Р	1 304,79 Р	499,17 Р	6 722,02 Р	11 777,98 Р
15	573,34 Р	8 600,15 Р	71,16 Р	1 375,95 Р	502,18 Р	7 224,20 Р	11 275,80 Р
16	573,34 Р	9 173,49 Р	68,12 Р	1 444,07 Р	505,22 Р	7 729,42 Р	10 770,58 Р

Рис. 15.5. График погашения ссуды



Рабочую книгу с примерами для данного раздела можно найти в файле `loan amortization schedule.xlsx` на сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется `Pic15_05.xlsx`.

Данные по взятой в банке ссуде (кредиту) введены в диапазон В1:В4. Они используются в формулах, содержащихся в ячейках, начиная с девятой строки. В табл. 15.2 представлены формулы из строки 9 графика погашения ссуды. Формулы были скопированы во все строки графика до 488 строки. Поэтому с помощью этого рабочего листа можно рассчитать график погашения ссуды продолжительностью до 480 периодов (40 лет).



Формулы в тех строках, которые не относятся к периодам выплат, будут возвращать ошибку. Поэтому для сокрытия данных в этих строках используются средства условного форматирования.




Подробнее об условном форматировании можно узнать в главе 21, “Визуализация данных средствами условного форматирования”.

**Таблица 15.2. Формулы для создания графика погашения ссуды**

Ячейка	Формула	Описание
A9	=A8+1	Возвращает номер платежа
B9	=ПЛТ(\$C\$2*(\$C\$3/12);\$C\$4;-\$C\$1)	Вычисляет размер периодического платежа
C9	=C8+B9	Возвращает общую (кумулятивную) сумму выплат
D9	=ПРПЛТ(\$C\$2*(\$C\$3/12);A9;\$C\$4;-\$C\$1)	Вычисляет процентную часть выплаты
E9	=E8+D9	Вычисляет кумулятивную сумму выплат по процентам
F9	=ОСПЛТ(\$C\$2*(\$C\$3/12);A9;\$C\$4;-\$C\$1)	Вычисляет размер тела кредита за одну выплату
G9	=G8+F9	Вычисляет кумулятивную сумму тела кредита
H9	=H8-F9	Вычисляет остаток суммы кредита в конце каждого периода

## Анализ параметров ссуды с помощью таблиц данных

Таблицы данных являются, по-видимому, одним из самых мощных средств Excel. Следует принять во внимание, что таблицы данных — это не совсем то же, что таблица (созданная с помощью команды Вставка⇒Таблицы⇒Таблица). Таблица данных — это удобное средство для подведения итогов вычислений, зависящих от одной или двух “изменяющихся” ячеек. В этом примере, изменяя значения одного или двух параметров ссуды, можно проследить их влияние на размер платежа. В следующих разделах описано, как создавать таблицы данных с одним или двумя входами.



В предыдущих русскоязычных версиях Excel таблицы данных назывались *таблицами подстановки*. — Примеч. ред.



Таблицы данных более подробно описаны в главе 35, “Анализ данных с помощью сценариев “что если””.



Рабочую книгу с примерами использования таблиц данных с одним и двумя входами можно найти в файле `loan data tables.xlsx` на веб-сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется `Pic15_06.xlsx`.

## Создание таблицы данных с одним входом

Таблица данных с одним входом позволяет выполнить произвольное количество расчетов, изменяя значения в одной ячейке.

На рис. 15.6 показана таблица (диапазон B10:I13), в которой вычисляются три параметра ссуды (размер периодического платежа, общая сумма выплат по кредиту и общая сумма выплат по процентам) при различных значениях процентной ставки, которая изменяется от 7,00 до 8,50% (всего семь значений). В данном примере значения подставляются в ячейку B2.

B6		=ПЛТ(B2*(B3/12);B4;-B1)							
	A	B	C	D	E	F	G	H	I
1	Размер ссуды:	100 000 Р							
2	Годовая процентная ставка:	7,25%							
3	Периодичность платежей (мес.):	1							
4	Количество периодов:	36							
5									
6	Размер платежа:	3 099 Р							
7	Сумма выплат:	111 570 Р							
8	Сумма процентов:	11 570 Р							
9									
10			7,00%	7,25%	7,50%	7,75%	8,00%	8,25%	8,50%
11	Размер платежа:	3 099 Р	3 088 Р	3 099 Р	3 111 Р	3 122 Р	3 134 Р	3 145 Р	3 157 Р
12	Сумма выплат:	111 570 Р	111 158 Р	111 570 Р	111 982 Р	112 396 Р	112 811 Р	113 227 Р	113 643 Р
13	Сумма процентов:	11 570 Р	11 158 Р	11 570 Р	11 982 Р	12 396 Р	12 811 Р	13 227 Р	13 643 Р
14									

**Рис. 15.6.** Пример таблицы данных с одним входом для вычисления параметров ссуды для различных значений процентной ставки

Чтобы создать таблицу данных с одним входом, выполните следующие действия.

- Введите формулы для вычисления параметров ссуды, которые будут использоваться в таблице данных.** В этом примере формулы находятся в диапазоне B6:B8.
- Введите значения изменяемой ячейки в соответствующие ячейки.** Здесь изменяемым значением является процентная ставка, значения которой находятся в ячейках C10:I10.
- В столбце, который находится слева от изменяемых значений, введите ссылки на ячейки с формулами.** В данном примере ссылки следует ввести в диапазон B11:B13. Например, в ячейке B11 содержится формула `=B6`.
- Выделите диапазон, который содержит все данные, созданные в предыдущих пунктах.** (В данном примере это диапазон B10:I13.)

5. **Выберите команду** Данные⇒Прогноз⇒Анализ “что если”⇒Таблица данных. На экране появится диалоговое окно, показанное на рис. 15.7.




Рис. 15.7. Диалоговое окно Таблица данных

6. **В поле Подставлять значения по столбцам в укажите ссылку на изменяемую ячейку.** В данном случае изменяемое значение хранится в ячейке B2.
7. **Второе поле оставьте пустым.** Оно используется для создания таблиц данных с двумя входами.
8. **Щелкните на кнопке ОК.** Excel создаст формулу массива, в которой используется функция ТАБЛИЦА с одним аргументом.
9. **По желанию отформатируйте полученную таблицу.** Например, можете добавить фон для заголовков таблицы.

Обратите внимание на то, что формула массива была введена не во все ячейки выделенного диапазона. Первые строка и столбец диапазона, выделенного в п. 4, остались без изменений.



Совет

При создании таблицы данных в левом столбце (в п. 3 в нем мы указали ссылки на ячейки с формулами) содержатся результаты вычислений для заданного в ячейке B2 значения изменяемого параметра. В нашем примере эти значения повторяются в столбце D. Поэтому, чтобы избежать недоразумений, лучше всего скрыть значения в левом столбце. Для этого можно, например, сделать одинаковыми цвет шрифта и цвет фона для левого столбца.

## Создание таблицы данных с двумя входами

Таблица данных с двумя входами содержит результаты вычислений по одной формуле при изменении двух входных параметров. На рис. 15.8 показана таблица данных с двумя входами (диапазон B10:I16), которая используется для вычисления размера платежа для семи значений процентной ставки и шести значений суммы кредита.

Чтобы создать таблицу данных с двумя входами, выполните такие действия.

1. **Создайте формулу, которая будет вычислять данные, используемые в таблице подстановки.** В этом примере такая формула находится в ячейке B6. Формулы в ячейках B7 и B8 не используются.
2. **Введите в строку значения первого изменяемого параметра.** В данном примере первым параметром является значение процентной ставки. Подставляемые значения хранятся в диапазоне C10:I10.
3. **Введите в столбец значения второго изменяемого параметра.** В этом примере размер ссуды выступает в качестве второго параметра, значения которого содержатся в диапазоне B11:B16.
4. **Создайте ссылку на ячейку, содержащую формулу для вычисления значений в таблице.** В нашем примере в ячейке B10 содержится ссылка =B6.

B6		=ПЛТ(B2*(B3/12);B4;-B1)								
	A	B	C	D	E	F	G	H	I	
1	Размер ссуды	100 000 Р								
2	Годовая процентная ставка	7,25%								
3	Периодичность платежей (мес.)	1								
4	Количество периодов	36								
5										
6	Размер платежа	3 099 Р								
7	Сумма выплат	111 570 Р								
8	Сумма процентов	11 570 Р								
9			<i>Годовая процентная ставка</i>							
10		3 099 Р	7,00%	7,25%	7,50%	7,75%	8,00%	8,25%	8,50%	
11	Размер ссуды	90 000 Р	2 779 Р	2 789 Р	2 800 Р	2 810 Р	2 820 Р	2 831 Р	2 841 Р	
12		95 000 Р	2 933 Р	2 944 Р	2 955 Р	2 966 Р	2 977 Р	2 988 Р	2 999 Р	
13		100 000 Р	3 088 Р	3 099 Р	3 111 Р	3 122 Р	3 134 Р	3 145 Р	3 157 Р	
14		105 000 Р	3 242 Р	3 254 Р	3 266 Р	3 278 Р	3 290 Р	3 302 Р	3 315 Р	
15		110 000 Р	3 396 Р	3 409 Р	3 422 Р	3 434 Р	3 447 Р	3 460 Р	3 472 Р	
16		115 000 Р	3 551 Р	3 564 Р	3 577 Р	3 590 Р	3 604 Р	3 617 Р	3 630 Р	

Рис. 15.8. Применение таблицы данных с двумя входами для вычисления размера платежа при разных параметрах ссуды

- Выделите диапазон, содержащий все ячейки**, в которые были введены данные на предыдущих шагах, в данном случае — диапазон В10:И16.
- Выберите команду Данные**⇒**Прогноз**⇒**Анализ “что–если”**⇒**Таблица данных**. Excel отобразит одноименное диалоговое окно, показанное на рис. 15.7.
- В поле Подставлять значения по столбцам в укажите ссылку на ячейку, которая содержит текущее значение первого параметра**. В данном примере следует ввести ссылку на ячейку В2.
- В поле Подставлять значения по строкам в введите ссылку на ячейку с текущим значением второго параметра таблицы подстановки**. В ашем примере создайте ссылку на ячейку В1.
- Щелкните на кнопке ОК**. Excel вставит формулу массива с функцией ТАБЛИЦА с двумя аргументами.

После создания таблицы данных можно изменить формулу, по которой выполняются расчеты. Для этого следует изменить ссылку в верхней левой ячейке таблицы подстановки. Например, если в этом примере заменить существующую ссылку в ячейке В10 ссылкой =В8, то в таблице будет рассчитываться общая сумма выплат по процентам, а если в ячейку В10 ввести ссылку =В7, то в таблице будет рассчитываться общая сумма выплат с учетом процентов.



Если таблица данных будет иметь очень большой размер, то скорость вычислений может существенно снизиться. Поэтому в Excel предусмотрен специальный режим вычислений для подобных случаев. Выберите команду **Формулы**⇒**Вычисление**⇒**Параметры вычислений**⇒**Автоматически**, кроме таблиц данных. После выбора этой команды пересчет таблиц данных будет выполняться не в автоматическом режиме, а после нажатия клавиши <F9>.

## Расчет параметров погашения ссуды нерегулярными платежами

До сих пор во всех примерах из этой главы рассчитывались параметры погашения ссуды (кредита) регулярными платежами. Но в некоторых случаях ссуда погашается нерегулярно. Например, вы заняли денег друзьям без официального соглашения о том, как будут производиться выплаты. И если при этом вас будет интересовать, какую сумму вы получите в качестве процентов, то в таком случае нужно выполнить вычисления на основании дат реальных платежей.

На рис. 15.9 представлен рабочий лист, с помощью которого можно отслеживать состояние таких займов. Значение годовой процентной ставки ссуды вводится в ячейку В1, которой присвоено имя ПС. Размер и дата первоначальной ссуды находятся в ячейках В5 и С5 соответственно. Формулы, расположенные в 6-й строке и ниже, позволяют отследить состояние ссуды, а также производят дополнительные расчеты.

A		B	C	D	E	F	G	H
1	Годовая процентная ставка (ПС):	5,00%	<b>История погашения ссуды</b>					
2								
3								
4	<b>Номер платежа</b>	<b>Размер платежа</b>	<b>Дата платежа</b>	<b>Проценты</b>	<b>Тело</b>	<b>Сумма платежей</b>	<b>Сумма по процентам</b>	<b>Баланс</b>
5	Основная ссуда	-7 500,00 Р	04.03.2014					7 500,00 Р
6	1	200,00 Р	20.04.2014	48,29 Р	151,71 Р	200,00 Р	48,29 Р	7 348,29 Р
7	2	200,00 Р	05.05.2014	15,10 Р	184,90 Р	400,00 Р	63,39 Р	7 163,39 Р
8	3	200,00 Р	20.06.2014	45,14 Р	154,86 Р	600,00 Р	108,53 Р	7 008,53 Р
9	4	100,00 Р	04.09.2014	72,97 Р	27,03 Р	700,00 Р	181,49 Р	6 981,49 Р
10	5	250,00 Р	15.10.2014	39,21 Р	210,79 Р	950,00 Р	220,70 Р	6 770,70 Р
11	Дополнительная ссуда	-500,00 Р	28.10.2014	12,06 Р	-512,06 Р	450,00 Р	232,76 Р	7 282,76 Р
12	6	100,00 Р	17.11.2014	19,95 Р	80,05 Р	550,00 Р	252,71 Р	7 202,71 Р
13	7	100,00 Р	23.11.2014	5,92 Р	94,08 Р	650,00 Р	258,63 Р	7 108,63 Р
14	8	1 000,00 Р	02.12.2014	8,76 Р	991,24 Р	1 650,00 Р	267,40 Р	6 117,40 Р
15	9	250,00 Р	17.12.2014	12,57 Р	237,43 Р	1 900,00 Р	279,97 Р	5 879,97 Р
16	10	200,00 Р	03.01.2015	13,69 Р	186,31 Р	2 100,00 Р	293,66 Р	5 693,66 Р
17	11	200,00 Р	20.01.2015	13,26 Р	186,74 Р	2 300,00 Р	306,92 Р	5 506,92 Р
18	12	1 000,00 Р	04.02.2015	11,32 Р	988,68 Р	3 300,00 Р	318,23 Р	4 518,23 Р
19	13	100,00 Р	16.02.2015	7,43 Р	92,57 Р	3 400,00 Р	325,66 Р	4 425,66 Р
20	14	200,00 Р	05.03.2015	10,31 Р	189,69 Р	3 600,00 Р	335,97 Р	4 235,97 Р
21	15	200,00 Р	22.03.2015	9,86 Р	190,14 Р	3 800,00 Р	345,83 Р	4 045,83 Р
22	16	100,00 Р	07.04.2015	8,87 Р	91,13 Р	3 900,00 Р	354,70 Р	3 954,70 Р
23	17	100,00 Р	17.04.2015	5,42 Р	94,58 Р	4 000,00 Р	360,12 Р	3 860,12 Р
24	Дополнительная ссуда	-500,00 Р	04.06.2015	25,38 Р	-525,38 Р	3 500,00 Р	385,50 Р	4 385,50 Р
25	18	100,00 Р	05.08.2015	37,25 Р	62,75 Р	3 600,00 Р	422,75 Р	4 322,75 Р
26	19	100,00 Р	18.08.2015	7,70 Р	92,30 Р	3 700,00 Р	430,44 Р	4 230,44 Р
27	20	200,00 Р	17.09.2015	17,39 Р	182,61 Р	3 900,00 Р	447,83 Р	4 047,83 Р
28	21	750,00 Р	08.10.2015	11,64 Р	738,36 Р	4 650,00 Р	459,47 Р	3 309,47 Р
29	22	750,00 Р	08.11.2015	14,05 Р	735,95 Р	5 400,00 Р	473,53 Р	2 573,53 Р
30	23	750,00 Р	06.12.2015	9,87 Р	740,13 Р	6 150,00 Р	483,40 Р	1 833,40 Р

**Рис. 15.9.** Рабочий лист, с помощью которого можно отследить параметры погашения ссуды нерегулярными платежами

В столбце В содержатся размеры платежей, сделанных в день, указанный в столбце С. Заметьте, что выплаты осуществлялись нерегулярно. Также следует отметить, что в двух случаях (строки 11 и 24) размер выплаты отрицательный.

тельный. Это значит, что производились дополнительные займы, которые добавлялись к остатку ссуды. Формулы в ячейках D и E вычисляют сумму выплат по процентам и телу кредита. Значения в столбцах F и G позволяют отследить общую сумму основных выплат и сумму выплат по процентам. Формулы в столбце H вычисляют остаток по ссуде после каждой выплаты. В табл. 15.3 приведены формулы, которые содержатся в строке 6, а также их краткое описание. Обратите внимание на то, что в каждой формуле используется функция ЕСЛИ, предназначенная для определения отсутствия даты платежа в столбце C. Если дата отсутствует, формула возвращает пустую строку и в ячейке, таким образом, данных не будет.

**Таблица 15.3. Формулы для вычисления параметров погашения ссуды нерегулярными платежами**

Ячейка	Формула	Описание
D6	=ЕСЛИ(C6<>""; (C6-C5)/365*H5*ПС; "")	По этой формуле вычисляется размер выплаты по процентам на основании даты платежа
E6	=ЕСЛИ(C6<>""; B6-D6; "")	Чтобы получить размер основной выплаты, из суммы платежа вычитается сумма выплат по процентам
F6	=ЕСЛИ(C6<>""; F5+B6; "")	Эта формула прибавляет сумму данного платежа к общей сумме выплат
G6	=ЕСЛИ(C6<>""; G5+D6; "")	Эта формула прибавляет сумму выплат по процентам к общей сумме
H6	=ЕСЛИ(C6<>""; H5-E6; "")	Формула вычисляет остаток в конце каждого периода



Рабочую книгу с примерами использования таблиц данных с примером вычисления параметров погашения ссуды нерегулярными платежами можно найти в файле `irregular payments.xlsx` на веб-сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется `Pic15_09.xlsx`.

## Расчет параметров инвестиций

При расчете параметров инвестиций основной интерес представляет вычисление прибыли от инвестиций с фиксированной процентной ставкой, таких как банковские сберегательные счета, депозиты или аннуитеты. Такого рода вычисления можно проводить для инвестиций, состоящих из одного или более вкладов.



Рабочую книгу со всеми расчетами, приведенными в этом разделе, можно найти в файле `investment calculations.xlsx` на веб-сайте книги ([www.wiley.com/go/excel-2016bible](http://www.wiley.com/go/excel-2016bible)). Для русскоязычного издания книги файл называется `Pic15_10.xlsx`.

## Будущая стоимость одного вклада

Большинство инвестиций представляет собой одноразовый вклад на определенный срок, проценты по которому выплачиваются по окончании этого срока. В настоящем разделе описаны способы вычисления простых и сложных процентов.

## Расчет выплат по простому проценту

В случае *простого процента* сумма выплаченных процентов не прибавляется к основной сумме вклада. Чтобы вычислить доход на основании простого процента, используется приведенная ниже формула:

Доход = Основная\_Сумма \* Процентная\_Ставка \* Срок

Например, предположим, что в банк на депозит вложено 10 000 Р на один год, при условии, что простая годовая процентная ставка составляет 5%. Через год банк вернет сумму вклада, а также выплатит проценты в размере 500 Р. Итого сумма выплаты составит 10 500 Р. В этом случае, чтобы вычислить полученные проценты, следует умножить основную сумму (10 000 Р) на процентную ставку (0,05) и на срок 1 год.

Если срок инвестиции меньше одного года, то простая процентная ставка будет соответствующим образом преобразована. Например, если в предыдущем примере срок вклада будет шесть месяцев, то по процентам банк выплатит только 250 Р, поскольку в этом случае годовая процентная ставка умножается на 6/12.

На рис. 15.10 показан рабочий лист, который содержит формулы, выполняющие расчет выплат по простому проценту. Формула в ячейке В7 (ей присвоено имя Доход) вычисляет размер выплат по процентам в конце срока:

=Основная\_Сумма\*Процентная\_Ставка\*Срок

По сути, она перемножает значения в ячейках В3 (Основная\_Сумма), В4 (Процентная\_Ставка) и В5 (Срок).

	A	B	C	D	E	F
1	<b>Расчет выплат по простому проценту</b>					
2						
3	Размер инвестиции:	10 000,00 Р				
4	Годовая процентная ставка:	5,00%				
5	Срок (лет):	1,00				
6						
7	Проценты:	500,00 Р				
8	Выплата в конце срока:	10 500,00 Р				
9						
10						

Рис. 15.10. Расчет выплат по простому проценту

Формула в ячейке В8 суммирует размер выплат по процентам и размер инвестиции.

## Расчет выплат по сложному проценту

В большинстве случаев выплаты по процентам вычисляются на основе одного из типов сложного процента. Под *сложным процентом* понимается многократное начисление процентов, которые добавляются к общей сумме вклада, причем в следующий раз проценты начисляются как по основной сумме вклада, так и по начисленным ранее процентам.



Например, предположим, что сумма депозита в банке составляет 10 000 Р. Годовая процентная ставка составляет 5%, проценты начисляются каждый месяц. После первого месяца проценты начисляются по основной сумме вклада, а затем полученная сумма выплат по процентам добавляется к основной. В следующем месяце проценты будут начисляться уже по большей сумме (основная плюс выплаты по процентам за первый месяц). Полученная сумма снова добавляется к текущей сумме вклада. Чтобы вычислить объем инвестиции в конце срока, можно создать последовательность формул (рис. 15.11).

Месяц	Проценты	Баланс
Начальная сумма		10 000,00 Р
1	41,67 Р	10 041,67 Р
2	41,84 Р	10 083,51 Р
3	42,01 Р	10 125,52 Р
4	42,19 Р	10 167,71 Р
5	42,37 Р	10 210,08 Р
6	42,54 Р	10 252,62 Р
7	42,72 Р	10 295,34 Р
8	42,90 Р	10 338,24 Р
9	43,08 Р	10 381,31 Р
10	43,26 Р	10 424,57 Р
11	43,44 Р	10 468,00 Р
12	43,62 Р	10 511,62 Р

Рис. 15.11. Чтобы выполнить расчет выплат по сложному проценту, можно создать последовательность формул

В столбце В содержатся формулы, вычисляющие выплаты по процентам для каждого месяца. Например, в ячейке В10 содержится следующая формула:

$$=C9*(\$B\$5*(1/12))$$

или

$$=C9*(\text{Процентная\_ставка}*(1/12))$$

Формулы в столбце С суммируют начисления по процентам за текущий месяц с суммой инвестиции за предыдущий месяц. Например, в ячейке С10 содержится такая формула:

$$=C9+B10$$

В конце срока (12 месяцев) сумма инвестиции составляет 10 511,62 Р. Таким образом, ежемесячное начисление сложного процента принесло (по сравнению с простым процентом) дополнительную прибыль в размере 11,62 Р.

Вместо того чтобы создавать последовательность формул, можно воспользоваться функцией БС. На рис. 15.12 показан рабочий лист, который содержит формулы для расчета выплат по сложному проценту. В ячейке В6 хранится число выплат за один год. Если выплаты производятся ежемесячно, то в ячейке В6 должно находиться число 12. Если выплаты будут производиться ежеквартально, то число выплат равно 4. Если же выплаты производятся ежедневно, то количество выплат будет равно 365. Срок инвестиции, указанный в виде количества лет, содержится в ячейке В7.

	A	B	C	D	E
1	<b>Расчет выплат по сложному проценту</b>				
2	<i>Вычисление с помощью одной формулы</i>				
3					
4	Размер инвестиций:	50 000,00 Р			
5	Годовая процентная ставка:	5,75%			
6	Количество выплат за год:	4			
7	Срок (лет):	3			
8					
9	Процентная ставка за период:	1,4375%			
10	Выплата в конце срока:	59 340,67 Р			
11	Проценты:	9 340,67 Р			
12					
13	Годовая доходность:	6,23%			
14					

Рис. 15.12. Расчет выплат по сложному проценту с помощью одной формулы

В ячейке В9 содержится приведенная ниже формула, которая вычисляет процентную ставку за один период. Полученное значение процентной ставки используется для вычисления суммы выплаты за каждый период:

=B5\*(1/B6)

или

=Процентная\_ставка\*(1/Выплат\_в\_году)

В формуле, которая хранится в ячейке В10, для расчета размера инвестиции в конце срока вклада используется функция БС:

=БС(В9;В6\*В7;;-В4)

или

=БС(Проценты\_за\_период;Выплат\_в\_году\*Срок;;-Основная\_сумма)

Первым аргументом функции БС является значение процентной ставки за один период, которое вычисляется в ячейке В9. Второй аргумент представляет собой общее число периодов. Третий аргумент пропущен, а четвертый аргумент — это размер инвестиции, который записан как отрицательное число.

Общая сумма выплат по процентам вычисляется с помощью простой формулы, записанной в ячейке В11:

=B10-B4

или

=Сумма\_выплат-Основная\_сумма

Формула в ячейке В13 позволяет вычислить годовую доходность (сумму начислений за год, выраженную в процентах относительно объема инвестиции):

= (В11/В4) / В7

или

= (Проценты/Основная\_сумма) /Срок

Например, пусть депозит в банке равен 50 000 Р, процентная ставка составляет 5,75% годовых, начисляемых ежеквартально. В этом случае количество периодов выплат за год равно 4 (ячейка В6). Срок вклада составляет три года, поэтому в ячейку В7 вводится число 3. Формула в ячейке В10 вернет значение 59 340,67 Р.

Предположим, что другой банк предлагает депозиты на тех же условиях, но начисления по процентам происходят ежедневно. На рис. 15.13 показаны параметры депозита размером 50 000 Р с ежедневными выплатами. Сравните полученные значения со значениями на рис. 15.12. Как видно, разница очень мала (общая сумма выплат по процентам равна 9340,67 и 9412,78 Р соответственно). За три года по второму вкладу будет выплачено на 72,11 Р больше, чем по первому. Если же сравнивать годовую доходность, то при ежеквартальных начислениях она будет составлять 6,23%, а при ежедневных — 6,28%.

Годовая_доходность		=(Проценты/Основная_сумма)/Срок	
A	B	C	D
1	<b>Расчет выплат по сложному проценту</b>		
2	<i>Вычисление с помощью одной формулы</i>		
3			
4	Размер инвестиций:	50 000,00 Р	
5	Годовая процентная ставка:	5,75%	
6	Количество выплат за год:	365	
7	Срок (лет):	3	
8			
9	Процентная ставка за период:	0,0158%	
10	Выплата в конце срока:	59 412,78 Р	
11	Проценты:	9 412,78 Р	
12			
13	Годовая доходность:	6,28%	

Рис. 15.13. Расчет выплат по сложному проценту с ежедневным начислением процентов

## Расчет выплат по сложному проценту с непрерывным начислением процентов

Термин *непрерывное начисление процентов* означает, что проценты начисляются и добавляются к сумме вклада постоянно. Иными словами, число периодов выплат по инвестиции за год — бесконечно. Следующая формула позволя-


ет определить будущую стоимость инвестиции объемом 50 000 Р, вложенную под 5,75% годовых, которые начисляются непрерывно на протяжении трех лет:

$=50000*EXP(0,0575*3)$

Эта формула вернет значение 59 413,59 Р, что на 0,08% больше по сравнению с аналогичной инвестицией, но с ежедневными выплатами (рис. 15.14).

	A	B	C
1	<b>Непрерывное начисление процентов</b>		
2			
3			
4	Размер инвестиций:	50 000,00 Р	
5	Годовая процентная ставка:	5,75%	
6	Начисление	Непрерывное	
7	Срок (лет):	3	
8			
9	Выплата в конце срока:	59 413,59 Р	
10			

**Рис. 15.14.** Расчет выплат по сложному проценту с непрерывным начислением процентов



Сложный процент можно вычислить без использования функции ЕС. Общая формула выглядит следующим образом:

$\text{сумма} * (1 + \text{ставка}) ^ \text{кпер}$

Здесь *сумма* — основная сумма инвестиции; *ставка* — процентная ставка за один период; *кпер* — общее количество периодов выплат.

Например, пусть инвестиция размером 50 000 Р вкладывается на пять лет под 5% годовых, которые начисляются ежемесячно. В этом случае будущую стоимость инвестиции можно вычислить по формуле

$=50000*(1+5\%/12)^(12*5)$

### Правило числа 72

Что делать, если нужно быстро принять решение о том, вкладывать деньги на определенных условиях или нет, а компьютера поблизости нет? Простое “правило числа 72” позволяет определить количество лет, по истечении которых стоимость инвестиции, вложенной под определенные проценты с ежегодными начислениями, удвоится. Просто разделите число 72 на значение процентной ставки. Например, пусть 100 000 Р вкладываются под 4% годовых. Через сколько лет эта сумма возрастет до 200 000 Р, т.е. удвоится? Разделив 72 на 4, получим, что стоимость данной инвестиции удвоится через 18 лет. А что будет, если процентная ставка будет равняться 5% годовых? В последнем случае сумма удвоится чуть больше чем через 14 лет.

Теперь рассмотрим точность этого правила. В таблице, приведенной ниже, во втором столбце представлены значения, полученные с помощью “правила числа 72”, а в правом столбце — реальные значения. Как видно, для малых значений процентной ставки это простое правило обеспечивает удивительную точность. Тем не менее помните, что, если процентная ставка будет равна 30% или больше, точность будет значительно меньше.

Процентная ставка	Правило числа 72	Фактическое значение
1%	72,00	69,66
2%	36,00	35,00
3%	24,00	23,45
4%	18,00	17,67
5%	14,40	14,21
6%	12,00	11,90
7%	10,29	10,24
8%	9,00	9,01
9%	8,00	8,04
10%	7,20	7,27
15%	4,80	4,96
20%	3,60	3,80
25%	2,88	3,11
30%	2,40	2,64

Правило числа 72 работает и в обратном направлении. Например, чтобы удвоить стоимость инвестиции за шесть лет, следует разделить 72 на 6. В результате получим, что деньги нужно вкладывать под 12% годовых. Остается пожелать успехов!

## Будущая стоимость серии вкладов

Рассмотрим еще один тип инвестиций, в котором на протяжении всего срока инвестиции следует периодически делать дополнительные вклады. Этот тип инвестиции называется *аннуитетом*. Некоторые банковские учреждения предлагают такую услугу.

Функции, описанные выше, в разделе “Кредитный калькулятор”, можно применить и для расчета аннуитетов, но для этого нужно изменить перспективу (кредитор вместо заемщика). Самый простой пример такого вида инвестиций — это целевые сберегательные вклады, предназначенные для проведения отпуска и предлагаемые целым рядом банковских организаций. Смысл состоит в том, что из каждой выплаты вам банк удерживает определенную фиксированную сумму, которая зачисляется в виде депозита на специальный сберегательный счет. Затем, через заранее оговоренный промежуток времени, например через один год, вы можете снять всю сумму (включая проценты) и использовать ее для проведения отпуска.

Предположим, в начале каждого месяца (на протяжении 12 месяцев) на счет вносится 2000 Р под 4,25% годовых, начисляемых ежемесячно. Следующая формула позволяет вычислить будущую стоимость депозита:

=БС(4,25%/12;12;-2000;;1)

Формула вернет значение 24 559,74 Р. Это число представляет собой сумму общего объема взносов (24 000 Р) и начислений по процентам (559,74 Р). Последний аргумент функции БС равен 1. Это значит, что взносы производятся в начале каждого периода. На рис. 15.15 показан рабочий лист, который содержит формулы для расчета параметров аннуитета. В табл. 15.4 представлены формулы этого листа с краткими описаниями.

		A	B	C	D	E	F	G
1	<b>Расчет параметров аннуитета</b>							
2								
3	<b>Депозиты...</b>							
4	Первый взнос:		0,00 Р					
5	Размер периодического взноса:		2 000,00 Р					
6	Число взносов за год:		12					
7	Взносы делаются в начале периода?		ИСТИНА					
8								
9	<b>Сроки инвестиций...</b>							
10	Срок инвестиций (лет):		1					
11								
12	<b>Процентная ставка...</b>							
13	Годовая процентная ставка:		4,25%					
14								
15	<b>Расчеты...</b>							
16	Первый взнос:		\$0,00					
17	Другие взносы:		\$24 000,00					
18	Всего внесено:		\$24 000,00					
19	Процентная ставка за период:		0,35%					
20	Сумма выплат в конце срока:		\$24 559,74					
21	Прибыль:		\$559,74					

Рис. 15.15. Расчет параметров аннуитета



Эту рабочую книгу можно найти в файле annuity calculator.xlsx на веб-сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется Pic15\_15.xlsx.

Таблица 15.4. Формулы для вычисления параметров аннуитета

Ячейка	Формула	Описание
B4	Значение вводится	Размер первого взноса (может быть равным нулю)
B5	Значение вводится	Размер периодического взноса
B6	Значение вводится	Число взносов за год
B7	Значение вводится	ИСТИНА, если взносы вносятся в начале периода; ЛОЖЬ — в противном случае
B10	Значение вводится	Срок инвестиции (лет). Значение может быть дробным
B13	Значение вводится	Значение годовой процентной ставки
B16	=B4	Отображает размер первого взноса
B17	=B5*B6*B10	Вычисляет общую сумму сделанных взносов
B18	=B16+B17	Вычисляет сумму первого и всех сделанных взносов
B19	=B13*(1/B6)	Вычисляет процентную ставку за период
B20	=БС(B19; B6*B10; -B5; -B4; ЕСЛИ(B7;1;0))	Вычисляет будущую стоимость инвестиции
B21	=B20-B18	Вычисляет прибыль от инвестиции

## Расчет амортизации

Программа Excel содержит пять функций, которые позволяют рассчитать амортизацию активов за время их эксплуатации. Значение амортизации в данный момент времени (уменьшение стоимости в заданный момент времени по сравнению с начальной стоимостью) можно описать с помощью функции, аргументами которой являются начальная стоимость и время эксплуатации. Выбор функции зависит от используемого *метода амортизации*.

В табл. 15.5 кратко описаны функции Excel, позволяющие рассчитать амортизацию, а также их аргументы. Дополнительную информацию можно найти в справочной системе Excel.

Таблица 15.5. Функции для расчета амортизации

Функция	Метод амортизации	Аргументы*
АПЛ	Линейный метод. Стоимость актива каждый год уменьшается на одинаковую сумму на протяжении всего срока эксплуатации	<i>нач_стоимость</i> ; <i>ост_стоимость</i> ; <i>время_эксплуатации</i>
ФУО	Метод фиксированного уменьшения остатка. Значение амортизации за период вычисляется с помощью фиксированной процентной ставки	<i>нач_стоимость</i> ; <i>ост_стоимость</i> ; <i>время_эксплуатации</i> ; <i>период</i> ; [ <i>месяцы</i> ]
ДДОВ	Метод двойного уменьшения остатка. Амортизация вычисляется с помощью увеличенного коэффициента. Значение амортизации максимально в первый период, а затем постепенно уменьшается	<i>нач_стоимость</i> ; <i>ост_стоимость</i> ; <i>время_эксплуатации</i> ; <i>период</i> ; [ <i>коэффициент</i> ]
АСЧ	Метод суммы годовых чисел. Устанавливает большее значение амортизации в первые периоды эксплуатации	<i>нач_стоимость</i> ; <i>ост_стоимость</i> ; <i>время_эксплуатации</i> ; <i>период</i>
ПУО	Метод снижающегося остатка. Вычисляет амортизацию для любого периода (включая частичные периоды) с использованием метода двойного уменьшения остатка или любого другого указанного метода	<i>нач_стоимость</i> ; <i>ост_стоимость</i> ; <i>время_эксплуатации</i> ; <i>нач_период</i> ; <i>кон_период</i> ; [ <i>коэффициент</i> ]; [ <i>без_переключения</i> ]

\* Аргументы в квадратных скобках задавать необязательно.

Для того чтобы лучше понять принцип работы функций для расчета амортизации, рассмотрим их аргументы.

- ***нач\_стоимость*** — начальная стоимость активов.
- ***ост\_стоимость*** — стоимость активов в конце срока эксплуатации.
- ***время\_эксплуатации*** — количество периодов, на протяжении которых эксплуатируются активы.
- ***период*** — период, по истечении которого производится расчет амортизации.
- ***месяцы*** — количество месяцев в первом году; если аргумент опущен, он полагается равным 12.
- ***нач\_период*** — начальный период для вычисления амортизации.
- ***кон\_период*** — конечный период для вычисления амортизации.

- **коэффициент** — процентная ставка снижающегося остатка; если аргумент опущен, то он полагается равным 2 (т.е. используется метод двойного уменьшения остатка).
- **без\_переключения** — логическое значение, которое указывает, следует ли использовать линейную амортизацию, если значение амортизации превышает величину, рассчитанную методом снижающегося остатка.

На рис. 15.16 приведен пример расчета амортизации с помощью функций АПЛ, ФУО, ДДОБ и ФСЧ. Начальная стоимость актива составляет 100 000 Р, срок эксплуатации — 10 лет, остаточная стоимость — 10 000 Р. В разделе, озаглавленном Значение амортизации, представлены ежегодные объемы амортизации актива, рассчитанные разными методами. Во втором разделе Стоимость актива представлены значения оцененной стоимости актива на протяжении всего срока его эксплуатации.

C8		=ФУО(\$B\$2;\$B\$4;\$B\$3;A8)			
	A	B	C	D	E
1	Актив:	Офисная мебель			
2	Начальная стоимость:	100 000,00 Р			
3	Срок эксл. (лет):	10			
4	Остаточная стоимость:	10 000,00 Р			
5					
6	<b>Значение амортизации</b>				
7	Год	АПЛ	ФУО	ДДОБ	АСЧ
8	1	9 000,00 Р	20 600,00 Р	20 000,00 Р	16 363,64 Р
9	2	9 000,00 Р	16 356,40 Р	16 000,00 Р	14 727,27 Р
10	3	9 000,00 Р	12 986,98 Р	12 800,00 Р	13 090,91 Р
11	4	9 000,00 Р	10 311,66 Р	10 240,00 Р	11 454,55 Р
12	5	9 000,00 Р	8 187,46 Р	8 192,00 Р	9 818,18 Р
13	6	9 000,00 Р	6 500,84 Р	6 553,60 Р	8 181,82 Р
14	7	9 000,00 Р	5 161,67 Р	5 242,88 Р	6 545,45 Р
15	8	9 000,00 Р	4 098,37 Р	4 194,30 Р	4 909,09 Р
16	9	9 000,00 Р	3 254,10 Р	3 355,44 Р	3 272,73 Р
17	10	9 000,00 Р	2 583,76 Р	2 684,35 Р	1 636,36 Р
18					
19					
20	<b>Стоимость актива</b>				
21	Год	АПЛ	ФУО	ДДОБ	АСЧ
22	0	100 000,00 Р	100 000,00 Р	100 000,00 Р	100 000,00 Р
23	1	91 000,00 Р	79 400,00 Р	80 000,00 Р	83 636,36 Р
24	2	82 000,00 Р	63 043,60 Р	64 000,00 Р	68 909,09 Р
25	3	73 000,00 Р	50 056,62 Р	51 200,00 Р	55 818,18 Р
26	4	64 000,00 Р	39 744,96 Р	40 960,00 Р	44 363,64 Р
27	5	55 000,00 Р	31 557,49 Р	32 768,00 Р	34 545,45 Р
28	6	46 000,00 Р	25 056,65 Р	26 214,40 Р	26 363,64 Р
29	7	37 000,00 Р	19 894,98 Р	20 971,52 Р	19 818,18 Р
30	8	28 000,00 Р	15 796,61 Р	16 777,22 Р	14 909,09 Р
31	9	19 000,00 Р	12 542,51 Р	13 421,77 Р	11 636,36 Р
32	10	10 000,00 Р	9 958,75 Р	10 737,42 Р	10 000,00 Р
33					

Рис. 15.16. Значения амортизации, полученные с использованием четырех функций Excel



Эту рабочую книгу можно найти в файле depreciation calculations.xlsx на веб-сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется Pic15\_16.xlsx.



На рис. 15.17 представлен график остаточной стоимости активов, вычисленный с использованием различных методов амортизации. Как видно, с использованием функции АПЛ стоимость актива снижается по прямой линии. При использовании других функций стоимость снижается по кривой, поскольку максимальное снижение происходит в первые периоды срока эксплуатации.





Рис. 15.17. Расчет остаточной стоимости активов с помощью разных методов амортизации

Функция ПУО используется в тех случаях, когда нужно вычислить амортизацию за несколько периодов (например, за второй и третий годы эксплуатации). На рис. 15.18 показан рабочий лист, который позволяет выполнить расчет амортизации с использованием функции ПУО. В ячейке В11 содержится формула  $=ПУО(B2;B4;B3;B6;B7;B8;B9)$

Она рассчитывает амортизацию актива за первые три года эксплуатации (начиная с периода 0 и заканчивая периодом 3).

## Финансовое прогнозирование

Под прогнозированием мы подразумеваем предсказание значений, основанное на «исторических» значениях (т.е. на значениях, которые отмечались в прошлом). Эти значения могут быть финансовыми (например, объемы продаж или величины дохода) или какими-либо другими данными, связанными со временем (например, количество работников).



В Excel 2016 включено пять новых функций рабочего листа для прогнозирования. Как видите, Excel может вставлять эти функции автоматически и может даже строить диаграмму.

B11		=ПУО(B2;B4;B3;B6;B7;B8;B9)			
A	B	C	D	E	F
1	Актив:	Офисная мебель			
2	Начальная стоимость:	100 000,00 Р			
3	Срок эксл. (лет):	10			
4	Остаточная стоимость:	10 000,00 Р			
5					
6	Начальный период:	0			
7	Конечный период:	3			
8	Коэффициент:	2			
9	Без переключения:	ИСТИНА			
10					
11	Амортизация:	48 800,00 Р			
12					

Рис. 15.18. Расчет амортизации за несколько периодов с помощью функции ПУО



Эту рабочую книгу можно найти в файле `forecasting example.xlsx` на веб-сайте книги ([www.wiley.com/go/excel2016bible](http://www.wiley.com/go/excel2016bible)). Для русскоязычного издания книги файл называется `Pic15_19.xlsx`.

Чтобы создать прогноз, начните с исторических, связанных со временем данных, например ежемесячных объемов продаж. На рис. 15.19 представлен простой пример. В столбце В содержатся данные ежемесячных объемов продаж с 2012 по 2015 год включительно. Кроме того, я построил график, из которого следует циклический характер изменения объемов продаж, причем продажи снижаются в летние месяцы. Цель заключается в том, чтобы предсказать ежемесячные объемы продаж на последующие два года.




Рис. 15.19. Ежемесячные объемы продаж за четыре года

Начните с выделения соответствующих данных. Для этого примера сначала выделите диапазон A1:В49. Затем выберите команду Данные⇒Прогноз⇒Лист прогноза, и Excel отобразит диалоговое окно Создание листа прогноза, показанное на рис. 15.20. (Я щелкнул на кнопке Параметры, чтобы отобразить на экране дополнительные параметры.) В этом диалоговом окне показан график, на котором отображаются “исторические” данные, прогнозируемые данные, а также доверительные интервалы для этого прогноза.

Доверительный интервал (на графике он показан более тонкими линиями) определяет “плюс/минус” значения для этого прогноза и указывает степень доверия к этому прогнозу. Повышая значение доверительного интервала, вы расширяете диапазон предсказания.

Обратите внимание: график, представленный в этом диалоговом окне, изменится при изменении вами параметров.




Рис. 15.20. Диалоговое окно Создание листа прогноза

Щелкните на кнопке Создать, и Excel вставит новый рабочий лист, который содержит таблицу и диаграмму. На рис. 15.21 представлена часть этой таблицы. В ней отображаются прогнозируемые значения, а также нижний и верхний дове-

рительные интервалы. Эти значения генерируются с помощью функций ПРЕДСКАЗ.ETS и ПРЕДСКАЗ.ETS.ДОВИНТЕРВАЛ. Это достаточно сложные функции (именно поэтому Excel берет на себя выполнение всей соответствующей работы).




Рис. 15.21. На рабочем листе прогноза содержатся таблица и график